# $Id: PKGBUILD 80564 2012-11-25 16:09:33Z kkeen $
# Maintainer: Kyle Keen <keenerd@gmail.com>
# Contributor: TDY <tdy@gmx.com>
# Contributor: DonVla <donvla@users.sourceforge.net>
# Contributor: Stefan Husmann <stefan-husmann@t-online.de>

pkgname=lilyterm
pkgver=0.9.9.2
pkgrel=1
pkgdesc="A light and easy to use libvte based X terminal emulator"
arch=('i686' 'x86_64')
url="http://lilyterm.luna.com.tw/index_en.html"
_watch="http://lilyterm.luna.com.tw/html/download.html"
license=('GPL3')
depends=('vte')
makedepends=('gettext' 'intltool' 'pkgconfig>=0.16')
source=(http://lilyterm.luna.com.tw/file/$pkgname-$pkgver.tar.gz)
md5sums=('d68dd29ce4894f765adba1efd470f90e')

build() {
  cd "$srcdir/lilyterm-$pkgver"
  ./configure --prefix=/usr \
              --sysconfdir=/etc
  make 
}

package() {
  cd "$srcdir/lilyterm-$pkgver"
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING \
    "$pkgdir/usr/share/licenses/lilyterm/COPYING"
}

